
Abstract

The anti-aliasing technique is something needed nowadays in com-
puter graphics to deal with under sampling problem. Generating an
image without any anti-aliasing technique is possible to see in many
situations ”jaggies” and ”moires”. This report has the intention of
showing some approaches using the screen space to solve the jaggy
problems, also called stair case problem. The Multisample Anti-
Aliasing is a technique that was widely used for real-time; how-
ever it is not compatible with some devices that use some screen-
space effects, such as deferred shading. Using the screen space
anti-aliasing approach it is possible to do the anti-aliasing in real-
time together with the screen space lighting effects. This report will
show how the Morphological Anti-aliasing (MLAA) and Subpixel
reconstruction anti-aliasing (SRAA) work and their implementation
in OpenGL/GLSL. The images showing the results are from the as-
signment proposed to implement those algorithms. Both techniques
get a image quality similar to MSAA between 4 and 8 samples and
performance of 0.44ms for MLAA ≈ 1ms for SRAA.

CR Categories: I.3.3 [Computer Graphics]: Three-Dimensional
Graphics and Realism—Realtime I.3.7 [Computer Graphics]:
Three-Dimensional Graphics and Realism—Anti-Aliasing;

1 Introduction

When generating images in computer graphics, without using any
technique to solve the aliasing problem, in many cases it is possi-
ble to see ”jaggies” and ”moires” . The jaggies are the staircase
effect in the edges and the moires are the unwanted pattern sample
[Hachisuka 2014]. Those both problems can be visualized in the
Figure 1.

Nowadays there are many techniques to solve those problems, in
the Figure 2 we can see the comparison of the results after using
the Multisample Anti-Aliasing (MSAA) technique.

As showed in the Figure 2, one way to solve this problem is us-
ing the Multisample Anti-Aliasing (MSAA), which basically you
take more samples per pixel and average during the generation of
the image. Up to now, many real-time graphics effects still using
the MSAA, however for some devices it is not possible to activate
multisampling and multiple render targets (MRT), that is crucial to
some technique as deferred shading [Jimenez et al. 2011]. In this
report, we are going to focus on how to solve the jaggies using the
screen space approach. The jaggy problem appear when it is not
possible to take more samples, in edges for example, as our screens
are made by pixels, it uses discretization of the real scene, so it

(a) Jaggy (b) Moire

Figure 1: The jaggies and moires that usually happen without any
anti-aliasing technique.

(a) Without anti-antialising (b) Using MSAA with 4 samples
per pixel

Figure 2: The difference between without and with anti-aliasing.



(a) Without anti-antialising (b) Using SSAA with 4 samples per
pixel

Figure 3: The difference between without and with anti-aliasing.

is needed to use some anti-aliasing techniques in order to render
smoother edges in order to generate high quality graphics.

Based on this limitation, some screen space techniques were devel-
oped in order to use the advantages of MRT with some anti-aliasing
technique. In this report it is going to be showed how the Mor-
phological Anti-Aliasing (MLAA) (Section 3) and Subpixel Re-
construction Anti-Aliasing (Section 4). A high level explanation of
MLAA is: detect the edges and measure the distance between pixels
of those edges to calculate the tonalization. The SRAA uses a high
resolution buffer of the geometry, so it can access subpixel infor-
mation in order to do the anti-aliasing. Both approaches represents
a quality between 4x - 8x MSAA and the performace of MLAA is
0.44 ms to render a 1280x720 resolution image using a NVIDIA
Geforce 9800GTX+ [Jimenez et al. 2011] and the performance of
SRAA is approximate 1 ms to render a 1280x720 resolution image
using a NVIDIA Geforce GTX480 [Chajdas et al. 2011].

2 Basic Anti-Aliasing techniques

In this section we are going to see how work some basic anti-
aliasing techniques that do not use the screen space effect. Those
techniques are SSAA (Section 2.1 and MSAA (Section 2.2).

2.1 Supersample Anti-Aliasing

The Supersample Anti-Aliasing is a simple idea to solve the alias-
ing problem. It simply render the image at higher resolution
and then down sample it using some filter, such as the box filter
[Hachisuka 2014]. During the down sample, it basically averages
the number of samples of each pixel. However it is a expensive ap-
proach, as it uses much memory. The Figure 3 show the comparison
without anti-aliasing and with anti-aliasing using this technique.

To do implementation of SSAA in OpenGL we need first to render
a higher resolution of the scene, as we can see in the Algorithm 1.
It just enables the multisample and creates a bigger image accord-
ing to the values of width and height. And then we just need to
downsample, as we can see in the Algorithm 2 written in GLSL. It
just takes the number of samples per pixel and get the average, 4
samples in the case of example showed in the Algorithm 2.

Algorithm 1: Render Scene to Multiple Sampled Texture

void Application::RenderSceneToMultisampledTexture()
{

// Enable multisampling!
glEnable(GL_MULTISAMPLE);
glEnable(GL_SAMPLE_SHADING);

// Bind the correct FBO!
glBindFramebuffer(GL_FRAMEBUFFER, m_uiMSFBOId);

// Set the viewport!
glViewport(0, 0, m_uiWidth, m_uiHeight);

// Do the rendering!
glEnable(GL_DEPTH_TEST);
glClearColor(0.0f, 0.5f, 1.0f, 0.0f);
glClear(GL_COLOR_BUFFER_BIT | GL_DEPTH_BUFFER_BIT);

RenderSceneShadedHelper(vec2(0.0f, 0.0f));

// disable multisampling!
glDisable(GL_MULTISAMPLE);

// unbind FBO
glBindFramebuffer(GL_FRAMEBUFFER, 0);

}

Algorithm 2: Downsample the high resolution

in FragData {
vec2 kTexcoord;

} IN;

out vec4 kOutputColor;

// Uniforms
uniform sampler2D kSceneTex;
uniform vec2 kHalfFineTexelSize;

void main()
{

vec4 samp1 = texture(kSceneTex, IN.kTexcoord +
vec2(-kHalfFineTexelSize.x,
-kHalfFineTexelSize.y));

vec4 samp2 = texture(kSceneTex, IN.kTexcoord +
vec2(-kHalfFineTexelSize.x,
+kHalfFineTexelSize.y));

vec4 samp3 = texture(kSceneTex, IN.kTexcoord +
vec2(+kHalfFineTexelSize.x,
-kHalfFineTexelSize.y));

vec4 samp4 = texture(kSceneTex, IN.kTexcoord +
vec2(+kHalfFineTexelSize.x,
+kHalfFineTexelSize.y));

//Average the samples
kOutputColor = (samp1 + samp2 + samp3 + samp4) / 4.0;

}

2.2 Multisample Anti-Aliasing

The Multisample Anti-Aliasing (MSAA) is one of the most pop-
ular anti-aliasing technique in real-time rendering [Jimenez et al.
2011]. Instead of getting the sample in the center of the pixel, it get
various offset samples per pixel, and average them to get the color
[Hachisuka 2014]. Normally 4 or 8 samples per pixel are used to
generate high quality images. It uses less memory comparing to
SSAA, because it uses the same resolution to do the anti-aliasing.
The Figure 2 show the result using this approach with 4 samples.

However, nowadays many games uses multiple render target (MRT)
technique, such as deferred shading to achieve better lighting ef-
fects, and MSAA and screen-space effects are not compatible. To
implement MSAA in OpenGL, we can just call glfwOpenWin-
dowHint(GLFW FSAA SAMPLES, 16), where the first parameter
is the type of anti-aliasing (sometimes MSAA is also called Full-



Screen Anti-Aliasing (FSAA)) and the second parameter is the
number of samples per pixel.

3 Morphological Anti-Aliasing

The Morphological Antialiasing (MLAA) is a technique that use the
multiple render targets (MRT) approach, so this technique is fully
compatible with modern lighting technique, such as deferred shad-
ing [Jimenez et al. 2011]. The MRT approach means, you store
various buffers, as such depth, normal, color, in order to do some
calculations. In a high level explanation, the MLAA find the edges
in the image and using this edge information, find for line ending,
then blend weights according to the line ending information and
finally do the anti-aliasing. The Figure 4 illustrates how it works.

Figure 4: High Level way of visualizing the MLAA algorithm.

As described, it uses 3 passes. Find edges (Section 3.1). Blend
weights calculation (Section 3.2). Neighborhood blending (Section
3.3). In which of those section will be explained how it is obtained
to do the MLAA.

3.1 Finding Edges

In order to find the edges, it is used the information from the depth
buffer. It simply measures the difference to the pixel on the top and
on the left [Jimenez et al. 2011] and use a threshold to decide if
it is a edge or not. However it is not capable of finding all edges.
The Figure 6 shows the edge detection using the depth information,
as we can see, comparing to the Figure 5, it does not capture the
edges related to the texture, so it is needed an approach that uses
the luminance in order to find the edges.

Figure 5: Original image.

In order to find all edges it is needed to get the luminance informa-
tion. The Equation 1 shows the way of computing the luminance,
where R,G and B are the red, green and blue channels for each pixel
of the image [Jimenez et al. 2011]. And it is also computed taking
the samples from top and left from the image and use a threshold

Figure 6: The edge detection using depth buffer.

number to decide if it is an edge. The Figure 7 show the result com-
bining the depth and luminance. As we can see, now it is possible
to get the edges related to the texture.

L = 0.2126 ∗R+ 0.7152 ∗G+ 0.0722 ∗B (1)

Figure 7: The edge detection using depth and luminance buffer.

This information about the edges computed is stored in another im-
age, and it will be the image that will go to the next pass. The
implementation in GLSL of edge detector using the depth and lu-
minance information can be seen in the Algorithm 3.

Algorithm 3: Edge Detector

void main() {
float fDepth = texture(kDepthTex, IN.kTexcoord).r;
float fDepthLeft = texture(kDepthTex, IN.kTexcoord

+ vec2(-1.0, 0.0) * kTexelSize).r;
float fDepthTop = texture(kDepthTex, IN.kTexcoord

+ vec2(0.0, 1.0) * kTexelSize).r;

vec2 kDepthDelta = abs(vec2(fDepth, fDepth) -
vec2(fDepthLeft, fDepthTop));

vec2 kEdges = step(vec2(fDepthThreshold,
fDepthThreshold), kDepthDelta);

float fLuminance = texture(kSceneTex,
IN.kTexcoord).r;

float fLuminanceLeft = texture(kSceneTex,
IN.kTexcoord + vec2(-1.0, 0.0) * kTexelSize).r;

float fLuminanceTop = texture(kSceneTex, IN.kTexcoord
+ vec2(0.0, 1.0) * kTexelSize).r;

vec2 kLuminanceDelta = abs(vec2(fLuminance,
fLuminance) - vec2(fLuminanceLeft,
fLuminanceTop));

kEdges += step(vec2(fLuminanceThreshold,
fLuminanceThreshold), kLuminanceDelta);

kOutputColor = kEdges;
}



3.2 Blend weights calculation

To calculate the blending weights, it is needed first for search for the
distance end line, using the edge information computed in the first
pass, then it is possible to fetch the crossing edges. The crossing
edges mean the type of pattern we are dealing with. In this pass we
also calculate the area according to the type of pattern. Using those
information computed we access the precalculated texture that we
store the areas to use to blend for the final pass.

To search for the distances to end lines, we perform an algorithm
that checks if we already reach the end of the line or not. The
Algorithm 4 shows how the search for end lines is computed from
right to left.

Algorithm 4: Edge Detector

float SearchXLeft(vec2 kTexcoord)
{

kTexcoord -= vec2(1.5, 0.0) * kTexelSize;

for (int i = 0; i < iMaxSearchSteps; ++i)
{

float fEdge = texture(kEdgesTex,
kTexcoord).g;

if (fEdge < 0.9)
return -2.0 * i - 2.0 * fEdge;

kTexcoord -= vec2(2.0, 0.0) * kTexelSize;
}

return -2.0 * iMaxSearchSteps;
}

3.3 Neighborhood blending

To blend the neighborhood we just need to blend it using the area
calculated in the previous pass. The Figure 8 illustrate the intuition
how it is calculated. The Equation 2 computes the new value of the
pixel, this is the last step of this algorithm, after computing the new
values on the edges, the anti-aliasing is done.

Figure 8: Neighborhood blending

cnew = (1− a) ∗ cold + a ∗ copp (2)

The Algorithm 5 shows how it is done in GLSL.

Algorithm 5: Neighborhood blending

void main()
{

// Look up weights for pixel at the top !

(a) Without anti-antialising (b) Result of MLAA

Figure 9: The difference between without and with MLAA.

float fTop = texture(kWeightsTex, IN.kTexcoord).r;
// Look up weights for pixel to the left !
float fLeft = texture(kWeightsTex, IN.kTexcoord).g;
// Look up weight for pixel at the bottom !
float fBottom = texture(kWeightsTex,

IN.kTexcoord).b;
// Look up weight for pixel to the right !
float fRight = texture(kWeightsTex,

IN.kTexcoord).a;
vec4 kWeights = vec4(fTop, fBottom, fLeft, fRight);
// Compute the sum of the weights
float fSum = dot(kWeights, vec4(1.0));
// If there is something to blend, do it
if (fSum > 0.0) {

// Convert the weights to texture offsets to
use bilinear filtering to do the
blending

vec4 kOffsets = kWeights * kTexelSize.yyxx;
vec4 kColor = vec4(0.0);
kColor += texture(kSceneTex, IN.kTexcoord +

vec2(0.0, kOffsets.r)) * kWeights.r;
kColor += texture(kSceneTex, IN.kTexcoord +

vec2(0.0, -kOffsets.g)) * kWeights.g;
kColor += texture(kSceneTex, IN.kTexcoord +

vec2(-kOffsets.b, 0.0)) * kWeights.b;
kColor += texture(kSceneTex, IN.kTexcoord +

vec2(kOffsets.a, 0.0)) * kWeights.a;
// Normalize the output
kOutputColor = kColor / fSum;

}
else {

kOutputColor = texture(kSceneTex,
IN.kTexcoord);

}
}

The final result of its implementation, can be visualized in the Fig-
ure 9.

4 Subpixel Reconstruction Anti-Aaliasing

The Subpixel Reconstruction Anti-Aliasing (SRAA) is a technique
that uses, as the MLAA, the screen space to do the anti-aliasing, so
it is compatible with lighting effects, such as deferred shading, that
uses the screen space to do the lighting. Different from the MLAA,
SRAA uses a higher resolution buffers of depth and normal in order
to do the anti-aliasing.

The steps of SRAA works as follow. First we generate buffers with
normals and positions information at a higher resolution in order to
get subpixel information. Second, we shade offset samples of the



scene. Third we use the SRAA filter that combines the processed
information from depth and normal with the shaded data.

The SRAA filter looks at each geometric sample in a pixel and anal-
yses all the surrounding shaded samples and computes a weight for
each. After that, the box filter is used to reconstruct color for geo-
metric sample.

To compute the weights of the pixels, it computes first the influence
of normal, given by the Equation 3. Then it takes the influence of
the depth buffer, given by the Equation 4. Using the information
from depth and normal, now we compute the weight that is given
by the Equation 5 where τ50.

dnormal = 1.0− dot(ngeo, nshade) (3)

ddepth = abs(dgeo − dshade) (4)

w = exp(−τ ∗max(dnormal, ddepth)) (5)

In order to compute the color reconstruction, we just weight the
neighboring shaded samples together, that is given by the Equations
6 and 7.

sum =

2∑
i=0

2∑
j=0

wi,j (6)

colorxy =

1∑
i=−1

1∑
j=−1

wi+1,j+1 ∗ colorx+i,j+i

sum
(7)

After that we just average the color to downsample using the box
filter. And it is how the SRAA works. It takes 1 ms to generate an
1280x720 image using a NVIDIA Geforce GTX 480.

To implement the SRAA, first we need to generate the offset. The
Algorithm 6 shows how it is generated. Then we take samples from
surrounding pixels in order to apply the SRAA filter, as it is showed
in the Algorithm 7. Then we need to apply the SRAA filter for
each sample. which is showed in the Algorithm 8. In the end we
reconstruct the color after the previous computation of the SRAA
filter. It can be seen in the Algorithm 9

Algorithm 6: Neighbourhood blending

// Collect the sample position of first sample !
float afFirstSamplePos[2];

glGetMultisamplefv(GL_SAMPLE_POSITION, 0,
&afFirstSamplePos[0]);

// Compute SRAA shading clip space offset !
m_afSRAAShadingOffset[0] = (0.5f -

afFirstSamplePos[0]) / ((float) m_uiWidth) *
2.0f;

m_afSRAAShadingOffset[1] = (0.5f -
afFirstSamplePos[1]) / ((float) m_uiHeight) *
2.0f;

Algorithm 7: Getting the information of the surrouding pixels

float afWeights[9];
float fSum = 0.0f;

ivec2 kCenterSample = ivec2(gl_FragCoord.xy);

/////////////////////////////////////
// First geometric sample in pixel //
/////////////////////////////////////
vec4 kNormalDepth = texelFetch(kNormalDepthTex,

kCenterSample, 0);
vec3 kNormal = kNormalDepth.xyz;

// Sample geometry of surrounding shaded samples
and compute weight for each

for (int j = 0; j < 3; j++)
{

for (int i = 0; i < 3; i++)
{

ivec2 kSample = kCenterSample +
ivec2(i-1, j-1);

// Sample tap
vec4 kSampleNormalDepth =

texelFetch(kNormalDepthTex,
kSample, 0);

// Compute weight
vec3 kSampleNormal =

kSampleNormalDepth.xyz;
float fW = Bilateral(kNormal,

kNormalDepth.w, kSampleNormal,
kSampleNormalDepth.w);

afWeights[i + j * 3] = fW;
fSum += fW;

}
}

Algorithm 8: SRAA filter

float Bilateral(vec3 kCenterNormal, float fCenterDepth,
vec3 kSampleNormal, float fTapDepth)

{
float dNormal = 1.0 - dot(kCenterNormal,

kSampleNormal);
float dDepth = abs(fCenterDepth - fTapDepth);
float T = 50.0;

return exp(-T * max(dNormal, dDepth));
}

Algorithm 9: Reconstructing the final color

// Reconstruct color for this sample
vec3 kReconstructedColor = vec3(0.0);

for (int j = -1; j < 2; j++)
{

for (int i = -1; i < 2; i++)
{

vec2 kSample = vec2(IN.kTexcoord.x + i

* kTexelSize.x, IN.kTexcoord.y +
j * kTexelSize.y);

kReconstructedColor += afWeights[(i+1)
+ (j+1) * 3] / fSum *
texture(kSceneTex, kSample).xyz;

}
}

The results of the SRAA can be seen in the Figure 10.



(a) Without anti-antialising (b) Result of SRAA

Figure 10: The difference between without and with SRAA.

5 Comparison with other topics

The techniques showed above, MLAA and SRAA, uses the screen-
space approach to do the anti-aliasing. The screen-space technique
is powerful in realtime rendering; we can use it for many lighting
effects, such as deferred shading and ambient occlusion.

The deferred shading decouples the geometry calculations (position
and normal transformations) and the lighting calculations. Instead
of shading each object, from the vertex shader, it gets the informa-
tion from all scene in order to shade it. So we get information from
the scene, such as position, normal, depth, color and store this in-
formation into a G-Buffer. With that information we can calculate
the light contribution for the whole scene. With many light sources,
deferred shading produces very realistic images and in realtime.

The ambient occlusion is a technique that measures how exposed
each point in a scene is to ambient lighting. It just generates many
sample points in a hemisphere na calculate if it is inside or outside
the objects. If there are many inside samples, it becomes darker
around the point, if there are many outside samples, it becomes
more white. It is a famous technique possible to generate a very
good visual effect and in real-time. Many movies and games use
this technique because of it is a cheap technique capable of creating
nice visual effects.

However those techniques were many times capable to use together
with the MSAA, because many devices do not support the activa-
tion of multiple rendering targets (MRT) and multisampling at the
same times [Jimenez et al. 2011]. So, screen-space anti-aliasing
was created to combine the powerful screen-space lighting with
anti-aliasing, generating images with high quality together with
beautiful lighting effects.

6 Conclusion

The MLAA and SRAA are very powerful techniques that is capable
of dealing with the aliasing problem generating smoother images
and combining with the screen-space lightinh effects. According
with the results showed by Jimenez et al., 2011 in MLAA and Cha-
jdas et al., 2011 in SRAA, the MLAA represents a faster algorithm,
as it works in the same resolution of the scene, the anti-aliasing is
calculated in less pixels, however the SRAA just applys the anti-
aliasing only the necessary parts of the scene, as we can see in the
Figure 11 [Chajdas et al. 2011]. The performance of MLAA is 0.44
ms to render a 1280x720 resolution image using a NVIDIA Geforce
9800GTX+ [Jimenez et al. 2011] and the performance of SRAA is
approximate 1 ms to render a 1280x720 resolution image using a
NVIDIA Geforce GTX480 [Chajdas et al. 2011]. Both approaches

represents a quality between 4x - 8x MSAA. However, comparing
all the results (Figure 12) with the scene of our implementation,
the MLAA has been slightly better. There are other techniques in
screen-space being developed, such as Enhanced Subpixel Morpho-
logical Antialiasing (SMAA) [Jimenez et al. 2012], this one is a
higher resolution version of the MLAA that shows better results
comparing to MLAA. There are always efforts in realtime render-
ing to create better algorithms, more efficient and with better visual
effects, and it always will be a great challenge in realtime computer
graphics.

Figure 11: Comparisons using another image.

(a) Without anti-antialising (b) Result of SSAA (c) Result of MSAA (d) Result of MLAA

(e) Result of SRAA

Figure 12: The visualization of all results.


